195
10-15-79

DR. 150

ORNL/CSD/TM-82

PIE—A FORTRAN Subroutine for Plotting Circles, Concentric Circles, Pie Charts, and Shaded Pie Charts

C. A. Dillard

MASTER

OAK RIDGE NATIONAL LABORATORY
OPERATED BY UNION CARBIDE CORPORATION · FOR THE DEPARTMENT OF ENERGY

Printed in the United States of America. Available from
National Technical Information Service
U.S. Department of Commerce
5285 Port Royal Road, Springfield, Virginia 22161
NTIS price codes—Printed Copy: A~~08~~; Microfiche A01
02

ORNL/CSD/TM-82
Dist. Category UC-32

Contract No. W-7405-eng-26

COMPUTER SCIENCES DIVISION

PIE — A FORTRAN SUBROUTINE FOR PLOTTING CIRCLES, CONCENTRIC CIRCLES, PIE CHARTS, AND SHADED PIE CHARTS

C. A. Dillard

(Sponsor: R. C. Durfee; Originator: C. A. Dillard)

Date Published - October 1979

NOTICE

This document contains information of preliminary nature. It is subject to revision or correction and therefore does not represent a final report.

UNION CARBIDE CORPORATION, NUCLEAR DIVISION
Operating the
Oak Ridge Gaseous Diffusion Plant . Oak Ridge National Laboratory
Oak Ridge Y-12 Plant . Paducah Gaseous Diffusion Plant
for the
DEPARTMENT OF ENERGY

CONTENTS

FIGURES

ACTIVITIES OF THE OAK RIDGE GEOGRAPHIC DATA SYSTEMS GROUP

Over the past seven years the Geographic Data Systems group has developed general-purpose computer techniques for a variety of environmental, land use, and energy related problems. The software development covers 1) data conversion and digitization, 2) data storage and retrieval, 3) data manipulation and analysis, and 4) graphical output and display. The software presently encompasses some 60 systems and subroutine packages utilizing large computers, minicomputers, digitizing systems and sophisticated graphics devices. Geographical data is handled in a variety of data structures including grids or rasters, point data, polygonal systems, triangulations, chains, and elemental line segments. The software/hardware configurations provide capabilities such as the digitization of maps and aerial photography, transformations between geographical data structures, rectification and interpolation of spatial data, data base management functions, integration of geographical entities and modeling, production of graphical displays in standard cartographic formats, etc. A few specific examples include routines to analyze DMA (Defense Mapping Agency) topographic data (e.g. slope, drainage patterns, aspect), determine land cover and regrid LANDSAT imagery data, transform map coordinates, convert cellular data to a polygonal basis, perform polygonal intersections, and plot perspective, shaded and contoured geographic displays. In cooperation with various sponsors, these capabilities have been combined for regional applications such as power plant siting, land-use modeling, coal and uranium resource analysis, strip mining impacts, aesthetic impacts of cooling towers, terrain modeling, water availability and consumption, waste disposal, and subsurface geological structure analysis. This group within the Computer Sciences Division is under the direction of Richard Durfee.

PIE — A FORTRAN SUBROUTINE FOR PLOTTING CIRCLES, CONCENTRIC CIRCLES, PIE CHARTS, AND SHADED PIE CHARTS

C. A. Dillard

ABSTRACT

This document describes a set of FORTRAN subroutines designed to draw circles, concentric circles, pie charts, and shaded pie charts. These subroutines will run on IBM/360 computers and the PDP-10. Subroutine PLOT is the only plotting routine used, allowing drawings to be made on the Versatec, Calcomp, or Tektronix.

I. INTRODUCTION

An IBM/360 FORTRAN subroutine, PIE, has been written to display information on the Calcomp or Versatec electrostatic plotter using circular diagrams. For example, in Fig. 1 a map of water availability and consumption is shown using shaded circles (also called shaded pie charts) to represent the data. The size of the circles is proportional to the millions of gallons per day available in the various drainage basins while the shaded portion indicates the percentages that will be consumed by all uses. The PIE subroutine was used to draw these circular patterns.

Four different types of patterns may be drawn: (1) simple circles, (2) concentric circles, (3) pie charts, and (4) shaded pie charts. Figure 2 shows examples of these four types. The first time PIE is called, it in turn calls a routine, CIRSET, to initialize all the parameters used in later calculations. Among these are two 37-element arrays, COSRAY



Fig. 1. Example Use of Subroutine PIE Showing Water Availability and Consumption



Fig. 2. The Four Types of Circles

and SINRAY, which contain precalculated sines and cosines to improve the calculating efficiency. A 360° circle is divided into 36 parts of 10° increments each. Since the first pie section begins with the vertical axis (north-south direction), the angles used in representing the circles range from 90° to 450° in a counterclockwise direction. Thus, the first pie section will be in the first quadrant rather than the fourth.

The circular patterns are always drawn about a center x,y origin with all parameters (radius, origin, etc.) given in plotter inches. Subroutine PLOT is the only basic plotting routine called by PIE, thus allowing the Calcomp, Tektronix, and the Versatec to be used. When drawing shaded pie charts, several trial runs may be needed to achieve the desired density of shading with the minimal number of concentric patterns. The use of an excessive number of shading lines is discouraged since it can greatly increase the execution time and plotting time. For darker shading, subroutine NEWPEN should be used to increase the line density on the Versatec and to change to a wider pen on the Calcomp. The routine can be changed to increase or decrease the number of circular sections (currently 36) to improve large plots or to cut down on execution time and plotting time.

There are several differences between subroutine PIE and DISSPLA's subroutine, APLPIE. APLPIE does not draw plain circles, concentric circles, or shaded pie charts. APLPIE draws its first sector on the X (horizontal) axis and inserts sectors counterclockwise. Subroutine PIE draws its first sector on the Y (vertical) axis and inserts sectors clockwise. If the summation of the percentages exceeds 100, APLPIE will not draw the pie chart; PIE will draw the chart stopping at 100%. PIE does not label the sectors, whereas APLPIE does.

II. SUBROUTINE PIE ARGUMENT LISTS

The user-supplied variables are REAL*4 or INTEGER*4 with examples using FORTRAN labeling standards (variables beginning with A–H and O–Z are real numbers, and variables beginning with I–N are integers).

Plain Circles

All variables in the argument list are supplied by the user. A model call to draw a plain circle is shown below with a description of the argument list. A sample plot (Fig. 3) is also shown.

```
CALL PIE(XCNTR,YCNTR,RADIUS,100.,RADIUS,1)
```

The variables are:

XCNTR,YCNTR = The X and Y coordinates of the center of the circle given in plotter inches.

RADIUS = The radius of the circle in plotter inches.



Fig. 3. Example Plot of a Plain Circle

Concentric Circles

The user supplies all the variables in the argument list. A model call, a description of the argument list, and an example figure (Fig. 4) are shown below.

```
CALL PIE(XCNTR,YCNTR,RADIUS,100.,SPAC,1)
```

The variables are:

XCNTR,YCNTR = The X and Y coordinates (in plotter inches) of the center of the circle.

RADIUS = The radius (in plotter inches) of the circle.

SPAC = The amount of spacing (in plotter inches) between the concentric circles. (Note: If the spacing (SPAC) is not exactly divisible into the radius (RADIUS), the smallest circle will have a radius smaller than SPAC.)



Fig. 4. Example Plot of Concentric Circles

Pie Charts

The user must supply all variables. A model call, an argument description, and an example figure (Fig. 5) are shown below.

```
CALL PIE(XCNTR,YCNTR,RADIUS,PERC,SPAC,NUM)
```

The variables are:

XCNTR,YCNTR = The X and Y coordinates of the center of the pie chart (in plotter inches).

RADIUS = The radius of the pie chart (in plotter inches).

PERC = A singly dimensioned real array whose elements are the percentage to be shown by each pie section.

SPAC = A singly dimensioned real array whose elements are set equal to RADIUS.

NUM = The number of pie sections to be drawn.



Fig. 5. Example Plot of a Pie Chart

The beginning side of the first sector starts on the vertical axis, and the sectors are inserted clockwise. It is not necessary that the summation of the elements of the PERC array equal 100. Subroutine PIE will automatically complete the circle. If the sum is greater than 100, the pie chart will be drawn stopping at 100%.

Shaded Pie Charts

The user supplies all variables in the argument list. Listed below is a model call and an argument description. An explanatory figure (Fig. 6) is also shown.

CALL PIE(XCNTR,YCNTR,RADIUS,PERC,SPAC,NUM)

The variables are:

XCNTR,YCNTR = The X and Y coordinates of the center of the pie chart (in plotter inches).

RADIUS = The radius of the pie chart (in plotter inches).

PERC = A singly dimensioned real array whose elements are the percentage to be shown by each pie section.

SPAC = A singly dimensioned real array whose elements are the amount of spacing (in plotter inches) between the shading lines in each pie section.

NUM = The number of pie sections to be drawn.



Fig. 6. Example Plot of a Shaded Pie Chart

III. EXAMPLE CALLS TO PIE

Circle

The call to draw a one-inch radius circle at (3.0, 2.0) is

```
CALL PIE (3.0, 2.0, 1.0, 100., 1.0, 1)
```

The plotter output is Fig. 7(a) on page 10.

Concentric Circles

The call to draw concentric circles 0.2 inches apart with a maximum radius of one inch at (3.0, 6.0) is

CALL PIE (3.0, 6.0, 1.0, 100., 0.2, 1)

The plotter output is Fig. 7(b) on page 10.

Pie Charts

To draw a pie chart with a one-inch radius at location (6.0, 3.0) and percentages 10.0% and 20.0%, the necessary call is

CALL PIE (6.0, 3.0, 1.0, PERC,SPAC,2)

with the elements of PERC being 10.0 and 20.0, and the elements of SPAC being 1.0 and 1.0. The plotter output is shown in Fig. 7(c) on page 10.

Shaded Pie Charts

The call to draw a shaded pie chart with a one-inch radius at location (6.0, 6.0), percentages of 8.0%, 12.0%, 15.0% and spacing of 0.2 inches, 0.1 inches, and 0.05 inches is

CALL (6.0, 6.0, 1.0, PERC,SPAC,3)

with array PERC set up as PERC(1) = 8.0, PERC(2) = 12.0, PERC(3) = 15.0, and array SPAC elements being 0.2, 0.1, and 0.05, respectively. The plotter output is Fig. 7(d) on page 10.

IV. USAGE AND APPLICATIONS

Estimate Time and Core Usages

Subroutine PIE and its companion subroutine CIRSET together require about 2 bytes of core. The CPU time on the IBM 360/91 to set up and draw the four figures on page 10 was 0.15 seconds. In setting up and drawing 100 concentric circles with 0.05 spacing and an average radius of 0.5 inch, the CPU time was 10.39 seconds.

ORNL-DWG 79-10691



Fig. 7. Output from Example Calls to PIE

Applications

Concentric circles generated by PIE were used on one plot to show the electrical plants. The size of the circle displayed the relative capacity.

In a wilderness area study, plain circles were drawn by PIE at such a scale that the size of the circle showed the actual acreage of a wilderness track located at the latitude, longitude centroid of the wilderness area.

A sample application of the use of shaded pie charts was given in the introduction.

APPENDIX A

FORTRAN LISTING

```
      SUBROUTINE CIRSET(COSRAY,SINRAY)
      DIMENSION COSRAY(37),SINRAY(37)
      DEGINC=10.0*(3.141592/180.)
      DO 10 I=1,36
      I1=I+8
      RAD=I1*DEGINC
      COSRAY(I)=-COS(RAD)
      SINRAY(I)=SIN(RAD)
10    CONTINUE
      COSRAY(37)=0.0
      SINRAY(37)=1.0
      RETURN
      END
```

```
      SUBROUTINE PIE(XCNTR,YCNTR,RADIUS,PERC,SPAC,NPERC)
      DIMENSION COSRAY(37),SINRAY(37)
      DIMENSION PERC(NPERC),SPAC(NPERC)
      LOGICAL FIRST
      DATA FIRST/.TRUE./
      IF(FIRST)CALL CIRSET(COSRAY,SINRAY)
      FIRST=.FALSE.
      CVRTR=3.141592654/180.
      ICOUNT=1
      XX=XCNTR+RADIUS*COSRAY(1)
      YY=YCNTR+RADIUS*SINRAY(1)
      OLDANG=90.0
      COSEND=0.0
      SINEND=1.0
      IF(NPERC.GT.1)CALL PLOT(XCNTR,YCNTR,3)
      IF(NPERC.GT.1)CALL PLOT(XX,YY,2)
      DO 20I=1,NPERC
      IF(PERC(I).EQ.0.0)GO TO 20
      ANG=PERC(I)*360./100.+OLDANG
      IF(ANG.GT.450.)ANG=450.
      ISTOP=INT(ANG/10.0)-8
      IF(ISTOP.GT.37)ISTOP=37
      RADENS=ANG*CVRTR
      J=0
      SHADER=RADIUS
      COSOLD=COSEND
      SINOLD=SINEND
      COSEND=-COS(RADENS)
      SINEND=SIN(RADENS)
30    J=J+1
      X=XCNTR+SHADER*COSOLD
      Y=YCNTR+SHADER*SINOLD
      CALL PLOT(X,Y,3)
      DO 40 K=ICOUNT,ISTOP
      X=XCNTR+SHADER*COSRAY(K)
      Y=YCNTR+SHADER*SINRAY(K)
      CALL PLOT(X,Y,2)
40    CONTINUE
      X=XCNTR+SHADER*COSEND
      Y=YCNTR+SHADER*SINEND
      CALL PLOT(X,Y,2)
      SHADER=SHADER-SPAC(I)
      IF(SHADER.LE.0)GO TO 50
      IF(J.LE.1000)GO TO 30
50    ICOUNT=ISTOP+1
      IF(PERC(I).GE.100.)     RETURN
      OLDANG=ANG
      CALL PLOT(XCNTR,YCNTR,3)
      XX = XCNTR+RADIUS*COSEND
      YY=YCNTR+RADIUS*SINEND
      CALL PLOT(XX,YY,2)
20    CONTINUE
      IF(ICOUNT.GT.37)  RETURN
      DO 60 JA=ICOUNT,37
      X=XCNTR+RADIUS*COSRAY(JA)
      Y=YCNTR+RADIUS*SINRAY(JA)
      CALL PLOT(X,Y,2)
60    CONTINUE
      RETURN
      END
```

ORNL/CSD/TM-82
Dist. Category UC-32

INTERNAL DISTRIBUTION

1. Donald Alvic
2. H. P. Carter/A. A. Brooks CSD Library
3. B. A. Clark
4. P. R. Coleman
5. C. A. Dillard
6. J. E. Dobson
7-41. R. C. Durfee
42. K. N. Fischer
43. J. F. Hines
44. P. E. Johnson
45. R. G. LeBlanc
46. S. M. Margle
47. K. W. Norton
48. S. K. Penny
49. P. R. Raby
50. A. D. Shepherd
51. E. P. Tinnel
52. G. W. Westley
53. D. L. Wilson
54-55. Central Research Library
56. Document Reference Section Y-12
57-59. Laboratory Records
60. Laboratory Records - Record Copy
61. ORNL Patent Office

EXTERNAL DISTRIBUTION

62. F. Eckerson, Grand Junction Office, Department of Energy, Box 2567, Grand Junction, Colorado 81501.
63. A. Voss, Tennessee Valley Authority, 216 Haney Building, 311 Broad Street, Chattanooga, Tennessee 37401.
64. Dr. F. Paul Baxter, Projects Manager, Natural Resources Analysis Staff, Division of Natural Resources Services, Tennessee Valley Authority, Norris, Tennessee 37828.
65. C. W. Smart, Supervisor, Division of Forestry, Fisheries & Wildlife, Tennessee Valley Authority, Norris, Tennessee 37828.
66. J. N. Rogers, Division 8324, Sandia Laboratories, Livermore, California 94550.
67. Chief, Mathematics and Geoscience Branch, Department of Energy Mathematical and Computer Sciences Program, Molecular Sciences and Energy Research, Division of Physical Research, Washington, D. C. 20545.
68. Office of Assistant Manager for Energy Research and Development Department of Energy, Oak Ridge Operations, Oak Ridge, Tennessee 37830.
69-259. Given distribution as shown in TID-4500 under UC-32 - Mathematics and Computers.